

Some Domain Theory and Denotational Semantics in Coq

Nick Benton¹, Andrew Kennedy¹, and Carsten Varming²

¹ Microsoft Research, Cambridge, UK
`{nick,akenn}@microsoft.com`

² Carnegie-Mellon University, Pittsburgh, USA
`varming@cmu.edu`

Abstract. We present a Coq formalization of constructive ω -cpo (extending earlier work by Paulin-Mohring) up to and including the inverse-limit construction of solutions to mixed-variance recursive domain equations, and the existence of invariant relations on those solutions. We then define operational and denotational semantics for both a simply-typed CBV language with recursion and an untyped CBV language, and establish soundness and adequacy results in each case.

1 Introduction

The use of proof assistants in formalizing language metatheory and implementing certified tools has grown enormously over the last five years or so. Most current work on mechanizing language definitions and type soundness results, certified compilation, proof carrying code, and so on has been based on operational semantics. But in our work on both certified compilation and on the semantics of languages with state, we have found ourselves wanting a Coq formalization of the kind of denotational semantics that we have grown accustomed to working with on paper.

Mechanizing domain theory and denotational semantics has an illustrious history. Provers such as HOL, Isabelle/HOL and Coq can all trace an ancestral line back to Milner's LCF [16], which was a proof checker for Scott's $PP\lambda$: a logic of cpo, continuous functions and admissible predicates. And although later systems were built on less domain-specific foundations, there have subsequently been dozens of formalizations of different notions of domains and bits of semantics, with examples in all the major provers. Few, however, have really gone far enough to be useful. This paper describes our Coq formalization of ω -cpo and the denotational semantics of both typed and untyped versions of a simple functional language, going considerably further than previous work. A companion paper [8] describes a non-trivial compiler correctness theorem that has been formalized and proved using one of these denotational models.

Our formalization is based on a Coq library for constructive pointed ω -cpo and continuous functions written by Paulin-Mohring [20] as a basis for a formal semantics of Kahn networks, and of probabilistic programs [6]. Section 2 describes our slight generalization of Paulin-Mohring's library to treat predomains

and a general lift monad. In Section 3, we then define a simply-typed call-by-value functional language, give it a denotational semantics using our predomains and prove the standard soundness and adequacy theorems, establishing the correspondence between the operational and denotational semantics. As far as we can tell, these results have not previously been mechanized for a higher-order language.

Section 4 is about solving recursive domain equations. We formalize Scott's inverse limit construction along the lines of later work by Freyd [11, 12] and Pitts [22, 23]. This approach characterizes the solutions as minimal invariants, yielding reasoning principles that allow one to construct and work with recursively-defined predicates and relations over the recursively-defined domains. In Section 5, we define the semantics of an untyped call-by-value language using a particular recursive domain, and use the associated reasoning principles to again establish soundness and adequacy theorems.

2 Basic Domain Theory

This first part of the development is essentially unchanged from the earlier work of Paulin-Mohring [20]. The main difference is that Paulin-Mohring formalized *pointed* cpos and continuous maps, with a special-case construction of flat cpos (those that arise from adding a bottom element under all elements of an otherwise discretely ordered set), whereas we use potentially bottomless cpos ('predomains') and formalize a general constructive lift monad.

2.1 Complete Partial Orders

We start by defining the type of preorders, comprising a carrier type *tord* (to which we can implicitly coerce), a binary relation *Ole* (written infix as \sqsubseteq), and proofs that *Ole* is reflexive and transitive:

```
Record ord: Type := mk_ord
  {tord :> Type;
   Ole : tord → tord → Prop;
   Ole_refl : ∀ x : tord, Ole x x;
   Ole_trans : ∀ x y z : tord, Ole x y → Ole y z → Ole x z}.
Infix "⊆" := Ole.
```

The equivalence relation $==$ is then defined to be the symmetrisation of \sqsubseteq :

```
Definition Oeq (O : ord) (x y : O) := x ⊆ y ∧ y ⊆ x.
Infix "==" := Oeq (at level 70).
```

Both $==$ and \sqsubseteq are declared as parametric **Setoid** relations, with \sqsubseteq being a partial order modulo $==$. Most of the constructions that follow are proved and declared to be morphisms with respect to these relations, which then allows convenient (in)equational rewriting in proofs.

The type of monotone functions between partial orders is a parameterized record type, comprising a function between the underlying types of the two order parameters and a proof that that function preserves order:

Definition *monotonic* ($O_1 O_2 : \text{ord}$) ($f : O_1 \rightarrow O_2$) :=
 $\forall x y, x \sqsubseteq y \rightarrow f x \sqsubseteq f y$.
Record *fmono* ($O_1 O_2 : \text{ord}$) : **Type** := *mk_fmono*
 $\{fmonot :> O_1 \rightarrow O_2;$
 $fmonotonic: monotonic fmonot\}$.

For any $O_1 O_2 : \text{ord}$, the monotonic function space $O_1 \rightarrow_m O_2 : \text{ord}$ is defined by equipping *fmono* $O_1 O_2$ with the order inherited from the codomain, $f \sqsubseteq g$ iff $f x \sqsubseteq g x$ for all x .

We define $\text{natO} : \text{ord}$ by equipping the set of natural numbers, nat , with the usual ‘vertical’ order, \leq . If $c : \text{natO} \rightarrow_m O$ for some $O : \text{ord}$, we call c a *chain* in O . Now a complete partial order is defined as a dependent record comprising an underlying order, *tord*, a function \sqcup for computing the least upper bound of any chain in *tord*, and proofs that this *is* both an upper bound (*le_lub*), and less than or equal to any other upper bound (*lub_le*):

Record *cpo*: **Type** := *mk_cpo*
 $\{tppo :> \text{ord};$
 $\sqcup : (\text{natO} \rightarrow_m tppo) \rightarrow tppo;$
 $le_lub : \forall (c : \text{natO} \rightarrow_m tppo) (n : \text{nat}), c n \sqsubseteq \sqcup c;$
 $lub_le : \forall (c : \text{natO} \rightarrow_m tppo) (x : tppo), (\forall n, c n \sqsubseteq x) \rightarrow \sqcup c \sqsubseteq x\}$.

This definition of a complete partial order is constructive in the sense that we require least upper bounds of chains not only to exist, but to be computable in Coq’s logic of total functions.

A monotone function f between two **cpo**s, D_1 and D_2 , is *continuous* if it preserves (up to $=$) least upper bounds. One direction of this is already a consequence of monotonicity, so we just have to specify the other:

Definition *continuous* ($D_1 D_2 : \text{cpo}$) ($f : D_1 \rightarrow_m D_2$) :=
 $\forall c : \text{natO} \rightarrow_m D_1, f (\sqcup c) \sqsubseteq \sqcup (f \circ c)$.

Record *fconti* ($D_1 D_2 : \text{cpo}$): **Type** := *mk_fconti*
 $\{fcontit : D_1 \rightarrow_m D_2;$
 $fcontinuous : continuous fcontit\}$.

For any $D_1 D_2 : \text{cpo}$, the continuous function space $D_1 \rightarrow_c D_2 : \text{ord}$ is defined by equipping the type *fconti* $D_1 D_2$ with the pointwise order inherited from D_2 . We then define $D_1 \Rightarrow_c D_2 : \text{cpo}$ by equipping $D_1 \rightarrow_c D_2$ with least upper bounds computed pointwise: if $c : \text{natO} \rightarrow_m (D_1 \rightarrow_c D_2)$ is a chain, then $\sqcup c : (D_1 \rightarrow_c D_2)$ is $\lambda d_1. \sqcup (\lambda n. c n d_1)$.

If $D : \text{cpo}$, write $\text{ID } D : D \rightarrow_c D$ for the continuous identity function on D . If $f : D \rightarrow_c E$ and $g : E \rightarrow_c F$ write $g \circ f : D \rightarrow_c F$ for their composition. Composition of continuous maps is associative, with ID as a unit.

Discrete cpos If $X : \text{Type}$ then equipping X with the order $x_1 \sqsubseteq x_2$ iff $x_1 = x_2$ (i.e. Leibniz equality) yields a **cpo** that we write **Discrete** X .

Finite products Write **1** for the one-point **cpo**, **Discrete unit**, which is terminal, in that for any $f g : D \rightarrow_c \mathbf{1}, f = g$. If $D_1 D_2 : \text{cpo}$ then equipping the usual product of the underlying types of their underlying orders with the pointwise ordering yields a product order. Equipping that order with a pointwise least upper bound operation $\sqcup c = (\sqcup (\text{fst} \circ c), \sqcup (\text{snd} \circ c))$ for $c \rightarrow_m D_1 \times D_2$ yields

a product cpo $D_1 \times D_2$ with continuous $\pi_i : D_1 \times D_2 \rightarrow_c D_i$. We write $\langle f, g \rangle$ for the unique (up to $==$) continuous function such that $f == \pi_1 \circ \langle f, g \rangle$ and $g == \pi_2 \circ \langle f, g \rangle$.

Closed structure We can define operations $\text{curry} : (D \times E \rightarrow_c F) \rightarrow (D \rightarrow_c E \Rightarrow_c F)$ and $\text{ev} : (E \Rightarrow_c D) \times E \rightarrow_c D$ such that for any $f : D \times E \rightarrow_c F$, $\text{curry}(f)$ is the unique continuous map such that $f == \text{ev} \circ \langle \text{curry } f \circ \pi_1, \pi_2 \rangle$. We define $\text{uncurry} : (D \Rightarrow_c E \Rightarrow_c F) \rightarrow_c D \times E \Rightarrow_c F$ by $\text{uncurry} = \text{curry}(\text{ev} \circ \langle \text{ev} \circ \langle \pi_1, \pi_1 \circ \pi_2 \rangle, \pi_2 \circ \pi_2 \rangle)$ and we check that $\text{uncurry}(\text{curry}(f)) == f$ and $\text{curry}(\text{uncurry}(h)) == h$ for all f and h .

So our internal category $\mathbb{CP}\mathbb{O}$ of cpos and continuous maps is Cartesian closed. We elide the details of other constructions, including finite coproducts, strict function spaces and general indexed products, that are in the formalization. Although our cpos are not required to have least elements, those that do are of special interest. We use Coq's typeclass mechanism to capture them:

```
Class Pointed(D : cpo) := {
  ⊥ : D;
  Pleast : ∀ d:D, ⊥ ⊆ d }.
Instance DOne_pointed : Pointed1.
Instance prod_pointed A B { pa : PointedA } { pb : PointedB } : Pointed(A × B).
Instance fun_pointed A B { pb : PointedB } : Pointed(A ⇒c B).
```

Now if D is `Pointed`, and $f : D \rightarrow_c D$ then we can define $\text{fix } f$, the least fixed point of f in the usual way, as the least upper bound of the chain of iterates of f starting at \perp . The proofs that this is well-defined and does yield a fixed point are just as on paper.

If $D : \text{cpo}$ and $P : D \rightarrow \text{Prop}$, then P is *admissible* if for all chains $c : \text{natO} \rightarrow_m D$ such that $(\forall n. P(c_n))$, one has $P(\bigsqcup c)$. In such a case, the subset type $\{d : D \mid P(d)\}$ with the order and lubs inherited from D is a cpo. We can also prove the standard fixed point induction principle:

```
Definition fix_ind D { pd : PointedD } : ∀ (F : D →m D) (P : D → Prop),
  admissible P → P ⊥ → (∀ x, P x → P (F x)) → P (fix F).
```

The main technical complexity in this part of the formalization is simply the layering of definitions, with (for example) `cpo`s being built on `ords`, and $D \Rightarrow_c E$ being built on $D \rightarrow_c E$, which is built on $D \rightarrow_m E$, which is built on $D \rightarrow E$. Definitions have to be built up in multiple staged versions and there are many implicit coercions and hints for Coq's `auto` tactic, which are tricky to get right. There is also much boilerplate associated with morphism declarations supporting setoid rewriting, and there is some tension between the elementwise and 'point-free' styles of working.

2.2 The lift monad

The basic order theory of the previous section goes through essentially as it does when working classically on paper. In particular, the definitions of lubs in products and function spaces are already constructive. But lifting will allow us to express general partial recursive functions, which, in Coq's logic of total

functions, is clearly going to involve some work. Our solution is a slight generalization of Paulin-Mohring’s treatment of the particular case of ‘flat’ cpos, which in turn builds on work of Capretta [9] on general recursion in type theory. We exploit Coq’s support for coinductive datatypes [10], defining lifting in terms of a type **Stream** of potentially infinite streams:

Variable $D : \text{cpo}$.

CoInductive $\text{Stream} : \text{Type} := \text{Eps} : \text{Stream} \rightarrow \text{Stream} \mid \text{Val} : D \rightarrow \text{Stream}$.

An element of **Stream** is (classically) either the infinite $\text{Eps}(\text{Eps}(\text{Eps}(\text{Eps}(\dots))))$, or some finite sequence of **Eps** steps, terminated by $\text{Val } d$ for some $d : D$, $\text{Eps}(\text{Eps}(\dots \text{Eps}(\text{Val } d) \dots))$. One can think of **Stream** as defining a resumptions monad, which we will subsequently quotient to define lifting. For $x : \text{Stream}$ and $n : \text{nat}$, $\text{pred_nth } x \ n$ is the stream that results from removing the first n **Eps** steps from x . The order on **Stream** is coinductively defined by

CoInductive $DLle : \text{Stream} \rightarrow \text{Stream} \rightarrow \text{Prop} :=$

| $DLle\text{Eps} : \forall x \ y, DLle \ x \ y \rightarrow DLle \ (\text{Eps } x) \ (\text{Eps } y)$
| $DLle\text{EpsVal} : \forall x \ d, DLle \ x \ (\text{Val } d) \rightarrow DLle \ (\text{Eps } x) \ (\text{Val } d)$
| $DLle\text{Val} : \forall d \ d' \ n \ y, \text{pred_nth } y \ n = \text{Val } d' \rightarrow d \sqsubseteq d' \rightarrow DLle \ (\text{Val } d) \ y$.


which satisfies the following coinduction principle:

Lemma $DLle_rec : \forall R : \text{Stream} \rightarrow \text{Stream} \rightarrow \text{Prop}$,

$(\forall x \ y, R \ (\text{Eps } x) \ (\text{Eps } y) \rightarrow R \ x \ y) \rightarrow$
 $(\forall x \ d, R \ (\text{Eps } x) \ (\text{Val } d) \rightarrow R \ x \ (\text{Val } d)) \rightarrow$
 $(\forall d \ y, R \ (\text{Val } d) \ y \rightarrow \exists n, \exists d', \text{pred_nth } y \ n = \text{Val } d' \wedge d \sqsubseteq d') \rightarrow$
 $\rightarrow \forall x \ y, R \ x \ y \rightarrow DLle \ x \ y$.

The coinduction principle is used to show that $DLle$ is reflexive and transitive, allowing us to construct a preorder $DL_ord : \text{ord}$ (and we now write the usual \sqsubseteq for the order). The infinite stream of **Eps**’s, Ω , is the least element of the order.

Constructing a **cpo** from DL_ord is slightly subtle. We need to define a function that maps chains $c : (\text{natO} \rightarrow_m DL_ord)$ to their lubs in DL_ord . An important observation is that if some c_n is non- Ω , i.e. there exists a d_n such that $c_n == \text{Val } d_n$, then for any $m \geq n$, there is a d_m such that $c_m == \text{Val } d_m$ and that moreover, the sequence d_n, d_{n+1}, \dots , forms a chain in D . Classically, the idea is that we look for such a c_n ; if we find one, then we can construct a chain in D and return Val applied to the least upper bound of that chain. If there’s no such chain then the least upper bound is Ω . But we cannot simply test whether a particular c_n is Ω or not: we can only examine finite prefixes. So we make a ‘parallel’ corecursive search through all the c_n s, which may be pictured something like this:³



The output we are trying to produce is an element of DL_ord . Each time our interleaving search finds an **Eps**, we produce an **Eps** on the output. So if every

³ In reality, the output stream ‘ticks’ less frequently than the picture would suggest.

element of the chain is Ω , we will end up producing Ω on the output. But should we find a $\text{Val } d$ after outputting some finite number of Eps s, then we know all later elements of the chain are also non- Ω , so we go ahead and build the chain in D that they form and compute its least upper bound using the lub operation of D .⁴ The details of this construction, and the proof that it does indeed yield the least upper bound of the chain c , involve interesting bits of constructive reasoning: going from knowing that there *is* a chain in D to actually having that chain in one's hand so as to take its lub uses (a provable form of) constructive indefinite description, for example. But at the end of the day, we end up with a constructive definition of $D_\perp : \text{cpo}$, which is clearly **Pointed**.

Lifting gives a strong monad [17] on $\mathbb{CP}\mathbb{O}$. The unit $\eta : D \rightarrow_c D_\perp$ applies the Val constructor. If $f : D \rightarrow_c E_\perp$ define $\text{kleisli } f : D_\perp \rightarrow_c E_\perp$ to be the map

$$\text{cofix } \text{kl } (d : D_\perp) : E_\perp := \text{match } d \text{ with } \text{Eps } dl \Rightarrow \text{Eps } (\text{kl } dl) \mid \text{Val } d' \Rightarrow f \ d'$$

Thinking operationally, the way in which kleisli sequences computations is very intuitive. To run $\text{kleisli } f \ d$, we start by running d . Every time d takes an Eps step, we do too, so if d diverges so does $\text{kleisli } f \ d$. Should d yield a value d' , however, the remaining steps are those of $f \ d'$. We prove that $\text{kleisli } f$ actually is a continuous function and, amongst other things, satisfies all the equations making $(-\perp, \eta, \text{kleisli}(-))$ a Kleisli triple on $\mathbb{CP}\mathbb{O}$. It is also convenient to have ‘parameterized’ versions of the Kleisli operators $\text{Kleisli } D \ E : (D \times E \rightarrow_c F_\perp) \rightarrow (D \times E_\perp \rightarrow_c F_\perp)$ defined by composing kleisli with the evident strength $\tau : D \times E_\perp \rightarrow_c (D \times E)_\perp$.

3 A Simply-Typed Functional Language

Our first application of the domain theory formalization is mechanize the denotational semantics of PCF_v , a simply-typed, call-by-value functional language with recursion. Types in PCF_v consist of integer, boolean, functions and products; we represent typing environments by a list of types.

Inductive $\text{Ty} := \text{Int} \mid \text{Bool} \mid \text{Arrow } (\tau_1 \ \tau_2 : \text{Ty}) \mid \text{Prod } (\tau_1 \ \tau_2 : \text{Ty})$.

Infix $" \rightarrow "$ $:= \text{Arrow}$. **Infix** $" * "$ $:= \text{Prod}$ (at level 55).

Definition $\text{Env} := \text{list } \text{Ty}$.

We separate syntactic values v from general expressions e , and restrict the syntax to ANF, with explicit sequencing of evaluation by LET and inclusion of values into expressions by VAL . As usual, there immediately arises the question of how to represent binders. Our first attempt used de Bruijn indices of type **nat** in a representation of syntax, and a separate type for typing judgments:

Inductive $\text{Value} := \text{VAR} : \text{nat} \rightarrow \text{Value} \mid \text{FIX} : \text{Ty} \rightarrow \text{Ty} \rightarrow \text{Exp} \rightarrow \text{Value} \mid \dots$

Inductive $\text{VTy } (\Gamma : \text{Env}) (\tau : \text{Ty}) : \text{Value} \rightarrow \text{Type} :=$

$\mid \text{TVAR} : \forall m, \text{nth_error } \Gamma m = \text{Some } \tau \rightarrow \text{VTy } \Gamma (\text{VAR } m) \ \tau$

$\mid \text{TFIX} : \forall e \ \tau_1 \ \tau_2, (\tau = \tau_1 \rightarrow \tau_2) \rightarrow \text{ETy } (\tau_1 :: \tau_1 \rightarrow \tau_2 :: \Gamma) \ e \ \tau_2 \rightarrow \text{VTy } \Gamma (\text{FIX } \tau_1 \ \tau_2 \ e) \ \tau \dots$

⁴ It doesn't matter which c_n we find a Val in first - the lub will be the same.

The major drawback of this approach is that typing judgments contain proof objects: simple equalities between types, in the case of *TFIX*, and a proof witnessing that a variable exists in the environment in the case of *TVAR*. This leads quickly into type-theoretic issues with proof irrelevance and decidable equality.

We recently switched to a strongly-typed term representation in which variable and term types are indexed by *Ty* and *Env*, ensuring that terms are well-typed by construction. This leads to concise and natural specification of the semantics and statements of theorems. However, it does test the power of Coq to an extreme, as dependencies are ubiquitous. Fortunately, new functionality in Coq 8.2, in particular ‘*Program*’ and ‘*dependent destruction*’, make for much simpler proofs compared with explicit manipulation of well-formedness conditions that was required in the previous approach.

We start by presenting the complete definition of well-typed terms.

```

Inductive Var : Env → Ty → Type :=
| ZVAR : ∀ Γ τ, Var (τ :: Γ) τ   | SVAR : ∀ Γ τ τ', Var Γ τ → Var (τ' :: Γ) τ.

Inductive Value : Env → Ty → Type :=
| TINT : ∀ Γ, nat → Value Γ Int   | TBOOL : ∀ Γ, bool → Value Γ Bool
| TVAR :> ∀ Γ τ, Var Γ τ → Value Γ τ
| TFIX : ∀ Γ τ1 τ2, Exp (τ1 :: τ2 :: Γ) τ2 → Value Γ (τ1 -> τ2)
| TPAIR : ∀ Γ τ1 τ2, Value Γ τ1 → Value Γ τ2 → Value Γ (τ1 * τ2)
with Exp : Env → Ty → Type :=
| TFST : ∀ Γ τ1 τ2, Value Γ (τ1 * τ2) → Exp Γ τ1
| TSND : ∀ Γ τ1 τ2, Value Γ (τ1 * τ2) → Exp Γ τ2
| TOP : ∀ Γ, (nat → nat → nat) → Value Γ Int → Value Γ Int → Exp Γ Int
| TGT : ∀ Γ, Value Γ Int → Value Γ Int → Exp Γ Bool
| TVAL : ∀ Γ τ, Value Γ τ → Exp Γ τ
| TLET : ∀ Γ τ1 τ2, Exp Γ τ1 → Exp (τ1 :: Γ) τ2 → Exp Γ τ2
| TAPP : ∀ Γ τ1 τ2, Value Γ (τ1 -> τ2) → Value Γ τ1 → Exp Γ τ2
| TIF : ∀ Γ τ, Value Γ Bool → Exp Γ τ → Exp Γ τ → Exp Γ τ.

Definition CExp τ := Exp nil τ. Definition CValue τ := Value nil τ.

```

Variables of type *Var* *Γ* *τ* are represented by a “typed” de Bruijn index that is in essence a proof that *τ* lives at that index in *Γ*. The typing rule associated with each term constructor can be read directly off its definition: for example, *TLET* takes an expression typed as *τ*₁ under *Γ*, and another expression typed as *τ*₂ under *Γ* extended with a new variable of type *τ*₁; its whole type is then *τ*₂ under *Γ*. The abbreviations *CExp* and *CValue* define closed terms.

Now the operational semantics can be presented very directly, as follows.

```

Inductive Ev : ∀ τ, CExp τ → CValue τ → Prop :=
| e_Val : ∀ τ (v : CValue τ), TVAL v ↓ v
| e_Op : ∀ op n1 n2, TOP op (TINT n1) (TINT n2) ↓ TINT (op n1 n2)
| e_Gt : ∀ n1 n2, TGT (TINT n1) (TINT n2) ↓ TBOOL (ble_nat n2 n1)
| e_Fst : ∀ τ1 τ2 (v1 : CValue τ1) (v2 : CValue τ2), TFST (TPAIR v1 v2) ↓ v1
| e_Snd : ∀ τ1 τ2 (v1 : CValue τ1) (v2 : CValue τ2), TSND (TPAIR v1 v2) ↓ v2
| e_App : ∀ τ1 τ2 e (v1 : CValue τ1) (v2 : CValue τ2),
  substExp [ v1, TFIX e ] e ↓ v2 → TAPP (TFIX e) v1 ↓ v2
| e_Let : ∀ τ1 τ2 e1 e2 (v1 : CValue τ1) (v2 : CValue τ2),
  e1 ↓ v1 → substExp [ v1 ] e2 ↓ v2 → TLET e1 e2 ↓ v2

```

$| e_IfTrue : \forall \tau (e_1 \ e_2 : \text{CExp } \tau) \ v, e_1 \Downarrow v \rightarrow TIF \ (TBOOL \ true) \ e_1 \ e_2 \Downarrow v$
 $| e_IfFalse : \forall \tau (e_1 \ e_2 : \text{CExp } \tau) \ v, e_2 \Downarrow v \rightarrow TIF \ (TBOOL \ false) \ e_1 \ e_2 \Downarrow v$
where "e '↓' v" := (Ev e v).

This relation makes use of a substitution operation. For example, in the e_Let rule, $substExp \ [v_1] \ e_2$ denotes the replacement in e_2 of variable with de Bruijn index 0 with value v_1 . Substitution on de Bruijn terms conventionally makes use of a *shift* operator, but the full dependent type of this operator (namely $Val \ (\Gamma \mapsto \Gamma') \ \tau \rightarrow Val \ (\Gamma \mapsto [\tau'] \mapsto \Gamma') \ \tau$) makes proofs very tricky in Coq's intensional type theory. A neat way around this issue is to define first a *renaming* operation, and use this to bootstrap a definition of substitution, defining *shift* in terms of renaming [5, 15, 1]. Definitions and lemmas regarding composition must be similarly bootstrapped: first composition of renamings is defined, then composition of substitution with renaming, and finally composition of substitutions.

3.1 Denotational semantics

The semantics of types is inductive, using product of cpo's to interpret products and continuous functions into a lifted cpo to represent call-by-value functions. The semantics of environments is simply defined pointwise, with environments building 'from the right' to simplify the semantics of fixpoint terms:

Fixpoint $SemTy \ \tau := \text{match } \tau \text{ with}$
 $| \text{Int} \Rightarrow Discrete \ \text{nat} \quad | \text{Bool} \Rightarrow Discrete \ \text{bool}$
 $| \tau_1 \rightarrow \tau_2 \Rightarrow SemTy \ \tau_1 \Rightarrow_c (SemTy \ \tau_2)_{\perp}$
 $| \tau_1 * \tau_2 \Rightarrow SemTy \ \tau_1 \times SemTy \ \tau_2 \quad \text{end.}$
Fixpoint $SemEnv \ \Gamma := \text{match } \Gamma \text{ with } nil \Rightarrow 1 \mid \tau :: \Gamma' \Rightarrow SemEnv \ \Gamma' \times SemTy \ \tau \text{ end.}$

We interpret $Value \ \Gamma \ \tau$ in $SemEnv \ \Gamma \rightarrow_c SemTy \ \tau$. Expressions are similar, except that the range is a lifted cpo. Note how we have used a 'point-free' style, with no explicit mention of value environments.

Fixpoint $SemVar \ \Gamma \ \tau \ (var : Var \ \Gamma \ \tau) : SemEnv \ \Gamma \rightarrow_c SemTy \ \tau :=$
 $\text{match } var \text{ with} \quad ZVAR \ _ \Rightarrow \pi_2 \quad | \ SVAR \ _ \ _ \ v \Rightarrow SemVar \ v \circ \pi_1 \quad \text{end.}$
Fixpoint $SemExp \ \Gamma \ \tau \ (e : Exp \ \Gamma \ \tau) : SemEnv \ \Gamma \rightarrow_c (SemTy \ \tau)_{\perp} :=$
 $\text{match } e \text{ with}$
 $| TOP \ _ \ op \ v_1 \ v_2 \Rightarrow \eta \circ \text{uncurry} \ (SimpleOp2 \ op) \circ \langle SemVal \ v_1, SemVal \ v_2 \rangle$
 $| TGT \ _ \ v_1 \ v_2 \Rightarrow \eta \circ \text{uncurry} \ (SimpleOp2 \ ble_nat) \circ \langle SemVal \ v_2, SemVal \ v_1 \rangle$
 $| TAPP \ _ \ _ \ v_1 \ v_2 \Rightarrow EV \circ \langle SemVal \ v_1, SemVal \ v_2 \rangle$
 $| TVAL \ _ \ _ \ v \Rightarrow \eta \circ SemVal \ v$
 $| TLET \ _ \ _ \ e_1 \ e_2 \Rightarrow Kleislr(SemExp \ e_2) \circ \langle ID, SemExp \ e_1 \rangle$
 $| TIF \ _ \ _ \ v \ e_1 \ e_2 \Rightarrow (choose \ _ \ @3 \ (SemExp \ e_1)) \ (SemExp \ e_2) \ (SemVal \ v)$
 $| TFST \ _ \ _ \ v \Rightarrow \eta \circ \pi_1 \circ SemVal \ v$
 $| TSND \ _ \ _ \ v \Rightarrow \eta \circ \pi_2 \circ SemVal \ v$
 $\text{end with } SemVal \ \Gamma \ \tau \ (v : Value \ \Gamma \ \tau) : SemEnv \ \Gamma \rightarrow_c SemTy \ \tau :=$
 $\text{match } v \text{ with}$
 $| TINT \ _ \ n \Rightarrow K \ _ \ (n : Discrete \ \text{nat})$
 $| TBOOL \ _ \ b \Rightarrow K \ _ \ (b : Discrete \ \text{bool})$
 $| TVAR \ _ \ _ \ i \Rightarrow SemVar \ i$
 $| TFIX \ _ \ _ \ e \Rightarrow FIXP \circ \text{curry} \ (\text{curry} \ (SemExp \ e))$
 $| TPAIR \ _ \ _ \ v_1 \ v_2 \Rightarrow \langle SemVal \ v_1, SemVal \ v_2 \rangle \quad \text{end.}$

3.2 Soundness and adequacy

We first prove *soundness*, showing that if an expression e evaluates to a value v , then the denotation of e is indeed the denotation of v . This requires that substitution commutes with the semantic meaning function. We define the ‘semantics’ of a substitution $s : \text{Subst } \Gamma' \Gamma$ to be a map in $\text{SemEnv } \Gamma \rightarrow_c \text{SemEnv } \Gamma'$.

Fixpoint $\text{SemSubst } \Gamma \Gamma' : \text{Subst } \Gamma' \Gamma \rightarrow \text{SemEnv } \Gamma \rightarrow_c \text{SemEnv } \Gamma' :=$
 $\text{match } \Gamma' \text{ with}$
 $\quad | \text{ nil } \Rightarrow \text{ fun } s \Rightarrow K _ (tt : \mathbf{1})$
 $\quad | _ :: _ \Rightarrow \text{ fun } s \Rightarrow \langle \text{SemSubst } (tlMap s) , \text{SemVal } (hdMap s) \rangle \text{ end.}$

This is then used to prove the substitution lemma, which in turn is used in the e_App and e_Let cases of the soundness proof.

Lemma $\text{SemCommutesWithSubst}$:

$(\forall \Gamma \tau (v : \text{Value } \Gamma \tau) \Gamma' (s : \text{Subst } \Gamma \Gamma'),$
 $\quad \text{SemVal } v \circ \text{SemSubst } s == \text{SemVal } (\text{substVal } s v))$
 $\wedge (\forall \Gamma \tau (e : \text{Exp } \Gamma \tau) \Gamma' (s : \text{Subst } \Gamma \Gamma'),$
 $\quad \text{SemExp } e \circ \text{SemSubst } s == \text{SemExp } (\text{substExp } s e)).$

Theorem Soundness : $\forall \tau (e : \text{CExp } \tau) v, e \Downarrow v \rightarrow \text{SemExp } e == \eta \circ \text{SemVal } v.$

We now prove *adequacy*, showing that if the denotation of a closed expression e is some (lifted) element, then e converges to a value. The proof uses a logical relation between syntax and semantics. We start by defining a liftRel operation that takes a relation between a cpo and values and lifts it to a relation between a lifted cpo and expressions, then use this to define relExp in terms of relVal .

Definition $\text{liftRel } \tau (R : \text{SemTy } \tau \rightarrow \text{CValue } \tau \rightarrow \text{Prop}) :=$
 $\text{fun } d \ e \Rightarrow \forall d', d == \text{Val } d' \rightarrow \exists v, e \Downarrow v \wedge R \ d' \ v.$

Fixpoint $\text{relVal } \tau : \text{SemTy } \tau \rightarrow \text{CValue } \tau \rightarrow \text{Prop} := \text{match } \tau \text{ with}$
 $\quad | \text{ Int } \Rightarrow \text{ fun } d \ v \Rightarrow v = \text{TINT } d$
 $\quad | \text{ Bool } \Rightarrow \text{ fun } d \ v \Rightarrow v = \text{TBOOL } d$
 $\quad | \tau_1 \rightarrow \tau_2 \Rightarrow \text{ fun } d \ v \Rightarrow \exists e, v = \text{TFIX } e \wedge \forall d_1 \ v_1, \text{relVal } \tau_1 \ d_1 \ v_1 \rightarrow \text{liftRel } (\text{relVal } \tau_2) (d \ d_1) (\text{substExp } [v_1, v] e)$
 $\quad | \tau_1 * \tau_2 \Rightarrow \text{ fun } d \ v \Rightarrow \exists v_1, \exists v_2, v = \text{TPAIR } v_1 \ v_2 \wedge \text{relVal } \tau_1 (\pi_1 d) v_1 \wedge \text{relVal } \tau_2 (\pi_2 d) v_2 \text{ end.}$

Fixpoint $\text{relEnv } \Gamma : \text{SemEnv } \Gamma \rightarrow \text{Subst } \Gamma \text{ nil} \rightarrow \text{Prop} := \text{match } \Gamma \text{ with}$
 $\quad | \text{ nil } \Rightarrow \text{ fun } _ \Rightarrow \text{True}$
 $\quad | \tau :: \Gamma \Rightarrow \text{ fun } d \ s \Rightarrow \text{relVal } \tau (\pi_2 d) (\text{hdMap } s) \wedge \text{relEnv } \Gamma (\pi_1 d) (tlMap s) \text{ end.}$

Definition $\text{relExp } \tau := \text{liftRel } (\text{relVal } \tau).$

The logical relation reflects $==$ and is admissible:

Lemma relVal_lower : $\forall \tau \ d \ d' \ v, d \sqsubseteq d' \rightarrow \text{relVal } \tau \ d' \ v \rightarrow \text{relVal } \tau \ d \ v.$

Lemma relVal_admissible : $\forall \tau \ v, \text{admissible } (\text{fun } d \Rightarrow \text{relVal } \tau \ d \ v).$

These lemmas are then used in the proof of the Fundamental Theorem for the logical relation, which is proved by induction on the structure of terms.

Theorem FT : $(\forall \Gamma \tau \ v \ \rho \ s, \text{relEnv } \Gamma \ \rho \ s \rightarrow \text{relVal } \tau (\text{SemVal } v \ \rho) (\text{substVal } s \ v))$
 $\wedge (\forall \Gamma \tau \ e \ \rho \ s, \text{relEnv } \Gamma \ \rho \ s \rightarrow \text{relExp } \tau (\text{SemExp } e \ \rho) (\text{substExp } s \ e)).$

Now we instantiate the fundamental theorem with closed expressions to obtain

Corollary Adequacy : $\forall \tau (e : \text{CExp } \tau) d, \text{SemExp } e \ tt == \text{Val } d \rightarrow \exists v, e \Downarrow v.$

4 Recursive Domain Equations

We now outline our formalization of the solution of mixed-variance recursive domain equations, such as arise in modelling untyped higher-order languages, languages with higher-typed store or languages with general recursive types.

The basic technology for solving domain equations is Scott's inverse limit construction, our formalization of which follows an approach due to Freyd [11, 12] and Pitts [23]. A key idea is to separate the positive and negative occurrences, specifying recursive domains as fixed points of locally continuous bi-functors $F : \mathbb{CPO}^{op} \times \mathbb{CPO} \rightarrow \mathbb{CPO}$, i.e. domains D such that $F(D, D) \simeq D$.

The type of mixed variance locally-continuous bifunctors on \mathbb{CPO} is defined as the type of records comprising an action on pairs of objects (*ob*), a continuous action on pairs of morphisms (*mor*), contravariant in the first argument and covariant in the second, together with proofs that *mor* respects both composition (*morph_comp*) and identities (*morph_id*):

```
Record BiFunctor : Type := mk_functor
{ ob : cpo → cpo → cpo;
  mor : ∀ (A B C D : cpo),
    (B ⇒c A) × (C ⇒c D) ⇒c (ob A C ⇒c ob B D) ;
  morph_comp : ∀ A B C D E F f g h k,
    mor B E D F (f.g) ∘ mor A B C D (h, k)
    == mor _ _ _ (h ∘ f, g ∘ k) ;
  morph_id : ∀ A B, mor _ _ _ (IDA, IDB) == ID_ }
```

We single out the *strict* bifunctors, taking pointed cpos to pointed cpos:

```
Definition FStrict : BiFunctor → Type :=
fun BF ⇒ ∀ D E, PointedD → PointedE → (Pointed(ob BF D E)).
```

We build interesting bifunctors from a few primitive ones in a combinatory style. If $D : \mathbf{cpo}$ then $BiConst D : \mathbf{BiFunctor}$ on objects is constantly D and on morphisms is constantly $ID D$. This is strict if D is pointed. $BiArrow : \mathbf{BiFunctor}$ on objects takes (D, E) to $D \Rightarrow_c E$ with the action on morphisms given by conjugation. This is strict. If $F : \mathbf{BiFunctor}$ then $BiLift F : \mathbf{BiFunctor}$ on objects takes (D, E) to $(ob F (D, E))_{\perp}$ and on morphisms composes *mor* F with the morphism part of the lift functor. This is always strict. If $F_1 F_2 : \mathbf{BiFunctor}$ then $BiPair F_1 F_2 : \mathbf{BiFunctor}$ on objects takes (D, E) to $(ob F_1 (D, E)) \times (ob F_2 (D, E))$ with the evident action on morphisms. This is strict if both F_1 and F_2 are. The definition of $BiSum F_1 F_2 : \mathbf{BiFunctor}$ is similar, though this is not generally strict as our coproduct is a separated sum.

A pair $(f : D \rightarrow_c E, g : E \rightarrow_c D)$ is an embedding-projection (e-p) pair if $g \circ f == id_D$ and $f \circ g \sqsubseteq id_E$. If F is a bifunctor and (f, g) an e-p pair, then $(mor F (g, f), mor F (f, g))$ is an e-p pair.

Now let $F : \mathbf{BiFunctor}$ and $FS : \mathbf{FStrict} F S$. We define $\langle D_i \rangle$ to be the sequence of **cpo**s defined by $D_0 = \mathbf{1}$ and $D_{n+1} = ob F (D_n, D_n)$. We then define a sequence of e-p pairs:

$$\begin{aligned} e_0 &= \perp : D_0 \rightarrow_c D_1 & p_0 &= \perp : D_1 \rightarrow_c D_0 \\ e_{n+1} &= mor F (p_n, e_n) : D_{n+1} \rightarrow_c D_{n+2} & p_{n+1} &= mor F (e_n, p_n) : D_{n+2} \rightarrow_c D_{n+1}. \end{aligned}$$

Let $\pi_i : \prod_j D_j \rightarrow_c D_i$ be the projections from the product of all the D_j s. The predicate $P : \prod_j D_j \rightarrow \mathbf{Prop}$ defined by $Pd := \forall i, \pi_i d == p_n(\pi_{i+1} d)$ is admissible, so we can define the sub-cpo D_∞ to be $\{d \mid Pd\}$ with order and lubs inherited from the indexed product. D_∞ will be the cpo we seek, so we now need to construct the required isomorphism.

Define $t_n : D_n \rightarrow D_\infty$ to be the map that for $i < n$ projects D_n to D_i via $p_i \circ \dots \circ p_{n-1}$ and for $i > n$ embeds D_n in D_i via $e_n \circ \dots \circ e_{i-1}$. Then $\text{mor } F(t_i, \pi_i) : \text{ob } F(D_\infty, D_\infty) \rightarrow_c \text{ob } F(D_i, D_i) = D_{i+1}$, so $t_{i+1} \circ \text{mor } F(t_i, \pi_i) : \text{ob } F(D_\infty, D_\infty) \rightarrow_c D_\infty$, and similarly $\text{mor } F(\pi_i, t_i) \circ \pi_{i+1} : D_\infty \rightarrow_c \text{ob } F(D_\infty, D_\infty)$. We then define

$$\begin{aligned} \text{UNROLL} &:= \bigsqcup_i (\text{mor } F(\pi_i, t_i) \circ \pi_{i+1}) : D_\infty \rightarrow_c \text{ob } F(D_\infty, D_\infty) \\ \text{ROLL} &:= \bigsqcup_i (t_{i+1} \circ \text{mor } F(t_i, \pi_i)) : \text{ob } F(D_\infty, D_\infty) \rightarrow_c D_\infty \end{aligned}$$

After some calculation, we show that $\text{ROLL} \circ \text{UNROLL} == \text{ID } D_\infty$ and $\text{UNROLL} \circ \text{ROLL} == \text{ID}(\text{ob } F(D_\infty, D_\infty))$, so we have constructed the desired isomorphism.

In order to do anything useful with recursively defined domains, we really need some general reasoning principles that allow us to avoid unpicking all the complex details of the construction above every time we want to prove something. One ‘partially’ abstract interface to the construction reveals that D_∞ comes equipped with a chain of retractions $\rho_i : D_\infty \rightarrow_c D_\infty$ such that $\bigsqcup_i \rho_i == \text{ID } D_\infty$; concretely, ρ_i can be taken to be $t_i \circ \pi_i$. A more abstract and useful principle is given Pitts’s [23] characterization of the solution as a *minimal invariant*, which is how we will establish the existence of a recursively defined logical relation in Section 5.1. Let $\delta : (D_\infty \Rightarrow_c D_\infty) \rightarrow_c (D_\infty \Rightarrow_c D_\infty)$ be given by

$$\delta e = \text{ROLL} \circ \text{mor } F(e, e) \circ \text{UNROLL}$$

The minimal invariance property is then the assertion that $\text{ID } D_\infty$ is equal to $\text{fix}(\delta)$, which we prove via a pointwise comparison of the chain of retractions whose lub we know to be the identity function with the chain whose lub gives the least fixed point of δ .

5 A Uni-Typed Lambda Calculus

We now apply the technology of the previous section to formalize the denotational semantics of an uni-typed (untyped) CBV lambda calculus with constants. As before, there are separate syntactic classes for values and expressions. This time the values are variables, numeric constants, and λ abstractions; expressions are again in ANF with *LET* and *VAL* constructs, together with function application, numeric operations, and a zero-test conditional. For binding, we use de Bruijn indices and a separate well-formedness judgment (we plan to move to a “strongly-typed” model in future). The evaluation relation is as follows.

Inductive Evaluation : $\mathbf{Exp} \rightarrow \mathbf{Value} \rightarrow \mathbf{Type} :=$

```

| e_Value : ∀ v, → VAL v ↓ v
| e_App : ∀ e v2 v, substExp [v2] e ↓ v → (APP (LAMBDA e) v2) ↓ v
| e_Let : ∀ e1 v1 e2 v2, e1 ↓ v1 → substExp [v1] e2 ↓ v2 → (LET e1 e2) ↓ v2
| e_lfz1 : ∀ e1 e2 v1, e1 ↓ v1 → IFZ (NUM O) e1 e2 ↓ v1
| e_lfz2 : ∀ e1 e2 v2 n, e2 ↓ v2 → IFZ (NUM (S n)) e1 e2 ↓ v2
| e_Op : ∀ op v1 n1 v2 n2, OP op (NUM n1) (NUM n2) ↓ NUM (op n1 n2)
where "e ↓ v" := (Evaluation e v).

```

5.1 Semantic Model

We interpret the untyped language in solution for the recursive domain equation $D \simeq (\mathbf{nat} + (D \rightarrow_c D))_{\perp}$, following the intuition that a computation either diverges or produces a value which is a number or a function. This is not the ‘tightest’ domain equation one could use for CBV: one could make function space strict, or equivalently make the argument of the function space be a domain of values rather than computations. But this equation still gives an adequate model. The construction in Coq is an instantiation of results from the previous section. First we build the strict bifunctor $F(D, E) = (\mathbf{nat} + (D \rightarrow_c E))_{\perp}$:

Definition FS := BiLift_strict (BiSum (BiConst (Discrete **nat**)) BiArrow).

And then we construct the solution, defining domains D_{∞} for computations and V_{∞} for values:

Definition D_{∞} := D_{∞} FS.

Definition V_{∞} := Dsum (Discrete **nat**) ($D_{\infty} \rightarrow_c D_{\infty}$).

Definition Roll : $(V_{\infty})_{\perp} \rightarrow_c D_{\infty}$:= ROLL FS.

Definition Unroll : $D_{\infty} \rightarrow_c (V_{\infty})_{\perp}$:= UNROLL FS.

Definition UR_iso : Unroll ◦ Roll == ID _ := DIsO_ur FS.

Definition RU_iso : Roll ◦ Unroll == ID _ := DIsO_ru FS.

For environments we define the n -ary product of V_{∞} and projection function.

Fixpoint SemEnv n : **cpo** := match n with 0 ⇒ 1 | $S\ n$ ⇒ SemEnv $n \times V_{\infty}$ end.

Fixpoint projenv $(m\ n:\mathbf{nat})$: $(m < n) \rightarrow \text{SemEnv } n \rightarrow_c V_{\infty}$:=

match m, n with

| $m, 0$ ⇒ fun inconsistent ⇒ match (lt_n_O m inconsistent) with end

| 0, $S\ n$ ⇒ fun _ ⇒ π_2

| $S\ m, S\ n$ ⇒ fun h ⇒ projenv (lt_S_n _ h) ◦ π_1 end.

We define a lifting operator $\text{Dlift} : (V_{\infty} \rightarrow_c D_{\infty}) \rightarrow_c D_{\infty} \rightarrow_c D_{\infty}$ and an application operator $\text{Dapp} : V_{\infty} \times V_{\infty} \rightarrow_c (V_{\infty})_{\perp}$ that given a pair looks at the first component; if it is a function then apply it to the second component, and if it is a natural number then diverge.

Definition Dlift : $(V_{\infty} \rightarrow_c D_{\infty}) \rightarrow_c D_{\infty} \rightarrow_c D_{\infty}$:=

curry (Roll ◦ ev ◦ (kleisli ◦ (Unroll ◦ -) ◦ π_1 , Unroll ◦ π_2)).

Definition Dapp : $V_{\infty} \times V_{\infty} \rightarrow_c (V_{\infty})_{\perp}$:=

ev ◦ ($\lfloor _ \rfloor : \text{Discrete } \mathbf{nat} \rightarrow_c D_{\infty} \rightarrow_c (V_{\infty})_{\perp}$, (Unroll ◦ -)) ◦ π_1 , Roll ◦ η ◦ π_2).

Now we can define the semantics of the language:

Fixpoint SemVal $v\ n$ ($vt : VTyping\ n\ v$) : SemEnv $n \rightarrow_c V_{\infty}$:=

match vt with

| TNUM n ⇒ INL _ ◦ (@K _ (Discrete **nat**) n)

```

| TVAR  $m \text{ } nthm \Rightarrow \text{projenv } nthm$ 
| TLAMBDA  $t \text{ } b \Rightarrow \text{INR } \_ \_ \circ \text{Dlift} \circ \text{curry} (\text{Roll} \circ \text{SemExp } b)$ 
end with SemExp  $e \text{ } n \text{ } (et : ETyping \text{ } n \text{ } e) : \text{SemEnv } n \rightarrow_c (\mathbf{V}_\infty)_\perp :=$ 
match  $et$  with
| TAPP  $\_ \_ v_1 \text{ } v_2 \Rightarrow \text{Dapp} \circ \langle \text{SemVal } v_1, \text{SemVal } v_2 \rangle$ 
| TVAL  $\_ \text{ } v \Rightarrow \eta \circ \text{SemVal } v$ 
| TLET  $\_ \_ e_1 \text{ } e_2 \Rightarrow \text{ev} \circ \langle \text{curry}(\text{Kleislr}(\text{SemExp } e_2)), \text{SemExp } e_1 \rangle$ 
| TOP  $op \text{ } \_ \_ v_1 \text{ } v_2 \Rightarrow$ 
    uncurry(Operator2 ( $\eta \circ \text{INL } \_ \_ \circ \text{uncurry}(\text{SimpleOp2 } op)$ ))  $\circ$ 
     $\langle [\eta, \perp : (\mathbf{D}_\infty \rightarrow_c \mathbf{D}_\infty) \rightarrow_c (\text{Discrete } \mathbf{nat})_\perp] \circ \text{SemVal } v_1,$ 
     $[\eta, \perp : (\mathbf{D}_\infty \rightarrow_c \mathbf{D}_\infty) \rightarrow_c (\text{Discrete } \mathbf{nat})_\perp] \circ \text{SemVal } v_2 \rangle$ 
| TIFZ  $\_ \_ \_ v \text{ } e_1 \text{ } e_2 \Rightarrow \text{ev} \circ$ 
     $\langle [[\mathbf{K} \_ (\text{SemExp } e_1), \mathbf{K} \_ (\text{SemExp } e_2)] \circ \text{zeroCase},$ 
     $\perp : (\mathbf{D}_\infty \rightarrow_c \mathbf{D}_\infty) \rightarrow_c \text{SemEnv } n \rightarrow_c (\mathbf{V}_\infty)_\perp] \circ (\text{SemVal } v), \text{ID } \_ \rangle \text{ end.}$ 

```

5.2 Soundness and Adequacy

As with the typed language, the proof of soundness makes use of a substitution lemma, and in addition uses the isomorphism of the domain \mathbf{D}_∞ in the case for *APP*. The proof then proceeds by induction, using equational reasoning to show that evaluation preserves semantics.

Lemma Soundness: $\forall e \text{ } v \text{ } (et : ETyping \text{ } 0 \text{ } e) \text{ } (vt : VTyping \text{ } 0 \text{ } v),$
 $e \Downarrow v \rightarrow \text{SemExp } et == \eta \circ \text{SemVal } vt.$

As in the simply typed case, adequacy is rather more difficult to prove than soundness. We again use a logical relation between syntax and semantics, but now we cannot define the relation by induction on types. Instead there will be a recursive specification of the logical relation over our recursively defined domain, and it is not at all clear that such a relation exists: because of the mixed variance of the function space, the operator on relations whose fixpoint we seek is not monotone. Following Pitts [23], however, we can again use the technique of separating positive and negative occurrences, defining a monotone operator in the complete lattice of *pairs* of relations, with the superset order in the first component and the subset order in the second. A fixed point of that operator can then be constructed by Knaster-Tarski.

We first define a notion of *admissibility* on relations between elements of our domain of values \mathbf{V}_∞ and syntactic values in **Value** and show that this is closed under intersection, so admissible relations form a complete lattice.

We then define a relational action corresponding to the bifunctor used in defining our recursive domain. This action, *RelF*, maps a pair of relations R, S on $(\mathbf{V}_\infty)_\perp \times \mathbf{Value}$ to a new relation T given by $(\text{inl } m, \text{NUM } m) \in T$ for all $m : \mathbf{nat}$ and if $f : \mathbf{D}_\infty \rightarrow_c \mathbf{D}_\infty$ is strict and for any $(d, v) \in R$ such that $0 \vdash v$ and any d' such that $\text{Unroll}(f(\text{Roll}(\text{Val } d))) == \text{Val } d'$ there exists a value v' and an evaluation $[[v]]e \Downarrow v'$ such that $(d', v') \in S$, then we have $(\text{inr } f, \text{LAMBDA } e) \in T$. A function in $\mathbf{D}_\infty \rightarrow_c \mathbf{D}_\infty$ is strict if and only if for all $v : (\mathbf{V}_\infty)_\perp$ and $v' : \mathbf{V}_\infty$, if $\text{Unroll}(f(\text{Roll}(v))) == \text{Val } v'$ then there exists a \hat{v} such that $v == \text{Val } \hat{v}$.

We then show that *RelF* maps admissible relations to admissible relations and is contravariant in its first argument and covariant in its second. Hence the

function $\lambda R : \text{RelAdm}^{op}. \lambda S : \text{RelAdm}. (\text{RelF } S \ R, \text{RelF } R \ S)$ is monotone on the complete lattice $\text{RelAdm}^{op} \times \text{RelAdm}$. Thus it has a least fixed point (Δ^-, Δ^+) . By applying the minimal invariant property from the previous section, we prove that in fact $\Delta^- == \Delta^+$, so we have found a fixed point, Δ of RelF , which is the logical relation we need to prove adequacy.

We extend Δ to Δ_e a relation on $(V_\infty)_\perp \times \mathbf{Exp}$ by $(d, e) \in \Delta_e$ if and only if for all d' if $d == \mathbf{Val} \ d'$ then there exist a value v and a derivation $e \Downarrow v$ such that $(d', v) \in \text{RelF } \Delta$.

The fundamental theorem for this logical relation is that for all environment env , derivations $n \vdash v$, and collection of derivations $ty : \Pi i \ v. \text{nth_error } vs \ i = \text{value } v \rightarrow n' \vdash v$, if $\text{nth_error } vs \ i = \text{value } v$ implies $(\text{projenv } i \ env, v) \in \Delta$ for all i and v then $(\llbracket n \vdash v \rrbracket, \llbracket vs \rrbracket v) \in \text{RelIV } \Delta \ \Delta$ and for all e , if $n \vdash e$ then $(\llbracket n \vdash e \rrbracket, \llbracket vs \rrbracket e) \in \Delta_e$.

Adequacy is then a corollary of the fundamental theorem:

Theorem Adequacy: $\forall e \ (te : \text{ETyping } 0 \ e) \ d \ \Gamma, \text{SemExp } te \ \Gamma \equiv \mathbf{Val} \ d \rightarrow \exists v, \exists ev : e \Downarrow v, \text{True}.$

6 Discussion

As we noted in the introduction, there have been many mechanized treatments of different aspects of domain theory and denotational semantics. One rough division of this previous work is between axiomatic approaches and those in which definitions and proofs of basic results about cpos, continuous functions and so on are made explicitly with the prover's logic. LCF falls into the first category, as does Reus's work on synthetic domain theory in LEGO [25]. In the second category, HOLCF, originally due to Regensburger [24] and later reworked by Müller et al [18], uses Isabelle's axiomatic type class mechanism to define and prove properties of domain-theoretic structures within higher order logic. HOLCPO [2, 4] was an extension of HOL with similar goals and basic definitions have also been formalized in PVS [7]. Coq's library includes a formalization by Kahn of some general theory of dcpos [14].

HOLCF is probably the most developed of these systems, and has been used to prove interesting results [19, 26], but working in a richer dependent type theory gives us some advantages. We can express the semantics of a typed language as a dependently typed map from syntax to semantics, rather than only being able to do shallow embeddings – this is clearly necessary if one wishes to prove theorems like adequacy or do compiler correctness. Secondly, it seems one really needs dependent types to work with conveniently with monads and logical relations, or to formalize the inverse limit construction.⁵

⁵ Agerholm [3] has formalized the construction of a model of the untyped lambda calculus using HOL-ST, an experimental version of HOL that also supports ZF-like set theory; this is elegant but HOL-ST is not widely used and no denotational semantics seems to have been done with it. Petersen has [21] formalized a reflexive cpo based on $P\omega$ in HOL, though this also appears not to have been developed far enough to be useful.

The constructive nature of our formalization and the coinductive treatment of lifting has both benefits and drawbacks. On the minus side, some of the proofs and constructions are much more complex than they would be classically and one does sometimes have to pay attention to which of two classically-equivalent forms of definition one works with (e.g. the definition of strictness in the previous section is the contrapositive of the ‘usual’ one). Worse, some constructions do not seem to be possible, such as the smash product of pointed domains; not being able to define \otimes was one motivation for our moving from Paulin-Mohring’s pointed cpos to our unpointed ones. One benefit that we have not so far mentioned, however, is that it is possible to extract actual executable code from the denotational semantics. Indeed, the lift monad can be seen as a kind of syntax-free operational semantics, not entirely unlike game semantics; this perspective, and possible connections with step-indexing, seem to merit further study.

The Coq development is of a fairly reasonable size. The domain theory library, including the theory of recursive domain equations, is around 7000 lines. The formalization of the typed language and its soundness and adequacy proofs are around 1700 lines and the untyped language takes around 2500. Although all the theorems go through (with no axioms), we have to admit that the development is still rather ‘rough’. Nevertheless, we have already used it as the basis of a non-trivial formalization of some new research [8] and our intention is to develop the formalization into something that is more widely useful. Apart from general polishing and refactoring, we are planning to abstract some of the structure of our category of domains, so as to make it convenient to work simultaneously with different categories at the same time, including categories of algebras. We would also like to provide some better support for ‘diagrammatic’ rewriting in monoidal (multi)categories. We currently find it very convenient to use Setoid rewriting to do pointfree equational reasoning that is a fairly direct translation of the normal commuting diagrams used in category theory. However, dealing with all the structural morphisms is still awkward and it should be possible to support something more like the diagrammatic proofs one can do with ‘string diagrams’ [13].

References

1. R. Adams. Formalized metatheory with terms represented by an indexed family of types. In *TYPES 2004*, volume 3839 of *LNCS*. Springer-Verlag, 2006.
2. S. Agerholm. Domain theory in HOL. In *Higher Order Logic Theorem Proving and its Applications*, volume 780 of *LNCS*, 1994.
3. S. Agerholm. Formalizing a model of the lambda calculus in HOL-ST. Technical Report 354, University of Cambridge Computer Laboratory, 1994.
4. S. Agerholm. LCF examples in HOL. *The Computer Journal*, 38(2), 1995.
5. T. Altenkirch and B. Reus. Monadic presentations of lambda terms using generalized inductive types. In *Computer Science Logic*, volume 1683 of *LNCS*, 1999.
6. P. Audebaud and C. Paulin-Mohring. Proofs of randomized algorithms in Coq. In *Mathematics of Program Construction*, volume 4014 of *LNCS*, 2006.
7. F. Bartels, A. Dold, H. Pfeifer, F. W. Von Henke, and H. Rue. Formalizing fixed-point theory in PVS. Technical report, Universitat Ulm, 1996.

8. N. Benton and C.-K. Hur. Logical relations for compiler correctness, 2009.
9. V. Capretta. General recursion via coinductive types. *Logical Methods in Computer Science*, 1, 2005.
10. T. Coquand. Infinite objects in type theory. In *Types for Proofs and Programs*, volume 806 of *LNCS*, 1993.
11. P. Freyd. Recursive types reduced to inductive types. In *Fifth Annual IEEE Symposium on Logic in Computer Science (LICS)*, 1990.
12. P. Freyd. Remarks on algebraically compact categories. In *Applications of Categories in Computer Science*, volume 177 of *LMS Lecture Notes*, 1992.
13. A. Joyal and R. Street. The geometry of tensor calculus. *Advances in Mathematics* 88, 1991.
14. G. Kahn. Elements of domain theory. In the Coq users' contributions library, 1993.
15. C. McBride. Type-preserving renaming and substitution. Unpublished draft.
16. R. Milner. Logic for computable functions: Description of a machine implementation. Technical Report STAN-CS-72-288, Stanford University, 1972.
17. E. Moggi. Notions of computation and monads. *Inf. Comput.*, 93(1):55–92, 1991.
18. O. Muller, T. Nipkow, D. von Oheimb, and O. Slotosch. HOLCF = HOL + LCF. *Journal of Functional Programming*, 9:191–223, 1999.
19. T. Nipkow. Winskel is (almost) right: Towards a mechanized semantics textbook. *Formal Aspects of Computing*, 10:171–186, 1998.
20. C. Paulin-Mohring. A constructive denotational semantics for Kahn networks in Coq. In *From Semantics to Computer Science. Essays in Honour of G Kahn*. 2009.
21. K. D. Petersen. Graph model of LAMBDA in higher order logic. In *Higher-Order Logic Users Group Workshop*, volume 780 of *LNCS*, 1993.
22. A. M. Pitts. Computational adequacy via 'mixed' inductive definitions. In *Mathematical Foundations of Programming Semantics*, volume 802 of *LNCS*, 1994.
23. A. M. Pitts. Relational properties of domains. *Information and Computation*, 127:66–90, 1996.
24. F. Regensburger. HOLCF: Higher order logic of computable functions. In *Theorem Proving in Higher Order Logics (TPHOLs)*, volume 971 of *LNCS*, 1995.
25. B. Reus. Formalizing a variant of synthetic domain theory. *Journal of Automated Reasoning*, 23:411–444, 1999.
26. C. Varming and L. Birkedal. Higher-order separation logic in Isabelle/HOLCF. In *Proceedings of MFPS 2008*, 2008.